This strangeness is caused by rounding errors and is not a paradox at all. If we compute results with exact numbers instead of approximate ones, we see both of your definitions of f(2) do indeed agree. I used Mathematica to compute these results.

As an example, let

$$f(1) = 4, f(n) = \frac{1}{nf(n+1)+1}$$

$$g(n) = 1/(1 + 2/(1 + 3/...1/(1 + nf(n + 1)))).$$

Observe $f(2) = g(n)$ for all values of n. Here we show this for $n = 5, 10, 20, 4000$:

```
In[205]:= (* compute the first 4001 values of f *)
          res = RecurrenceTable[{f[n] == 1 / (1 + n f[n + 1]), f[1] == 4}, f[n], {n, 4001}];

In[206]:= g[n_] := ContinuedFractionK[Piecewise[{{k f[k + 1], k == n}, {k, True}}], 1, {k, 1, n}]

In[207]:= g[5]
```

Out[207]= $\dfrac{1}{1+\dfrac{2}{1+\dfrac{3}{1+\dfrac{4}{1+5\,f[6]}}}}$

```
In[208]:= g[10]
```

Out[208]= $\dfrac{1}{1+\dfrac{2}{1+\dfrac{3}{1+\dfrac{4}{1+\dfrac{5}{1+\dfrac{6}{1+\dfrac{7}{1+\dfrac{8}{1+\dfrac{9}{1+10\,f[11]}}}}}}}}}$

```
In[209]:= res[[2]]
```

Out[209]= $-\frac{3}{4}$

```
In[210]:= {g[5], g[10], g[20], g[4000]} /. f[n_] :> res[[n]]
```

Out[210]= $\left\{-\frac{3}{4}, -\frac{3}{4}, -\frac{3}{4}, -\frac{3}{4}\right\}$

Here, the variable *res* holds the first 4001 values of f and $g(n)$ computes the n-long expansion of $f(2)$. Note $f(2)$ equals all values of g.

Now the number 0.525135276... arises from rounding error. If we round values of f, we introduce error (call it ϵ) and get an expression of the form

$$g_\epsilon(5) = \frac{1}{1 + \frac{2}{1+\frac{3}{1+\frac{4}{1+5(f(6)+\epsilon)}}}}.$$

Now when we collapse this fraction, it turns out the coefficient in the numerator and denominator of ϵ match the numerator and denominator of the convergents of the continued fraction mentioned in the problem statement.

If we let $g(n) = (a_1 + a_2 f(n+1))/(b_1 + b_2 f(n+1))$, then g_ϵ is of the form

$$g_\epsilon(n) = \frac{a_1 + a_2 f(n+1)}{b_1 + b_2 f(n+1) + b_2\epsilon} + \frac{a_2\epsilon}{b_1 + b_2 f(n+1) + b_2\epsilon},$$

where the coefficients are functions of n. The first term in the sum resembles $g(n)$, however the extra term in the denominator makes it tends to 0 as $n \to \infty$. Additionally, the second term tends to a_2/b_2 as $n \to \infty$, and a_2/b_2 tends to 0.525135276...